[image: image1.png]3 ASP.NEY Portal - Microsoft Internet Explorer

e Gt fpote Do 190
s moipponpsrost

ASP.NET Portal Starter Kit

T oty s product o biscussons About th portal

Welcome to the Portal Starter Kit This Week's Special

ASPNET
STARTER
R

Upcoming Events Top Movers

Catedory  (Sians)
TR oy svotanss o el w e





       Whitepapers

ASP.NET Starter Kit: Portal
November 2002

© 2002 Microsoft Corp. All rights reserved.

The information contained in this document represents the current view of Microsoft Corp. on the issues discussed as of the date of publication. Because Microsoft must respond to changing market conditions, it should not be interpreted to be a commitment on the part of Microsoft, and Microsoft cannot guarantee the accuracy of any information presented after the date of publication.

This white paper is for informational purposes only. MICROSOFT MAKES NO WARRANTIES, EXPRESS OR IMPLIED, AS TO THE INFORMATION IN THIS DOCUMENT.

Complying with all applicable copyright laws is the responsibility of the user. Without limiting the rights under copyright, no part of this document may be reproduced, stored in or introduced into a retrieval system, or transmitted in any form or by any means (electronic, mechanical, photocopying, recording or otherwise), or for any purpose, without the express written permission of Microsoft.

Microsoft may have patents, patent applications, trademarks, copyrights or other intellectual property rights covering subject matter in these documents. Except as expressly provided in any written license agreement from Microsoft, the furnishing of this document does not give any license to these patents, trademarks, copyrights or other intellectual property.

Microsoft, Visual Studio, Windows, IntelliSense, Visual Basic, Visual C#, MSDN, Windows NT, and JScript are either registered trademarks or trademarks of Microsoft Corp. in the United States and/or other countries.

The names of actual companies and products mentioned herein may be the trademarks of their respective owners.

Microsoft Corp. • One Microsoft Way • Redmond, WA 98052-6399 • USA

Summary: This article describes the design and architecture decisions for the ASP.NET Portal Starter Kit. In addition, a detailed look and explanation of the code required to extend it is also covered. (1 printed pages)

Overview

What is the Portal Starter Kit?

The Portal Starter Kit demonstrates how you can use ASP.NET along with the Microsoft .NET Framework to build intranet and Internet portal applications. The sample shows off many key features available with ASP.NET and also provides a “best practices” application that developers can use as a base to build their own ASP.NET applications.

The portal demonstrates many features offered by the ASP.NET technology including:

· Cross-browser support for Netscape and Internet Explorer 

· Mobile device support for WAP/WML and Pocket Browser devices 

· Clean code/html content separation using server controls 

· Pages that are constructed from dynamically-loaded user controls 

· Configurable output caching of portal page regions 

· Multi-tier application architecture

· ADO.NET data access using SQL stored procedures 

· Windows authentication - username/password in Active DS or NT SAM 

· Forms authentication using a database for usernames/passwords 

· Role-based security to control user access to portal content

This white paper discusses the portal in depth and provides insight from the perspective of the creators. In addition, the article covers how the portal can be used as a template for building online portals by examining many of the key application features and the technology used to implement them. 

The Portal Starter Kit was developed in both in-line and code behind versions. The SDK version is coded in-line and is optimized for the ASP.NET Web Matrix Project and the .NET Framework SDK. The second version was written using Microsoft ® Visual Studio.NET™ using the code-behind coding model. Both versions contain implementations written in C# and VB.NET. 

Application Architecture 

The portal uses a multi-tier application architecture.  The Portal contains two data sources. The configuration settings are stored in PortalCFG.xml and the content for the application is stored in a SQL Server database.  The data access is provided through a Microsoft .NET assembly that provides access to the data source via the stored procedures.  In addition, the portal framework is built through the use of a number of assemblies that handle the security and configuration of the portal.  Web Forms and user controls make up the presentation layer and handle the display and management of the portal data for the user.

Database

All of the content for the portal is stored in a SQL Server database. This allows server administrators to farm the front-end of the portal across a number of servers each pulling from a single unique data store. This section provides an overview of the database used in the portal. 

Database Schema

The portal database schema is a very simple schema that contains a simple repository for Module content and 3 tables to store Users and User Roles. The physical schema is shown in Figure 1.

Figure 1. Physical Database Schema

Portal Configuration XML Schema

The schema based on the PortalCFG.xml file contains all the configuration settings for the Portal. The schema is simple and easy to understand. The XML Configuration file stores all the high level Portal, Tab and Module Definitions. The configuration settings are stored in a cache and GetSiteSettings() only reads from the xml file if the settings have changed. The physical schema is shown in Figure 2.

Figure 2. PortalCFG.xsd Schema

Stored Procedures

The portal uses stored procedures to encapsulate all of the database queries. Stored procedures provide a clean separation between the database and the middle-tier data access layer. This in turn provides easier maintenance, since changes to the database schema will be invisible to the data access components. Using stored procedures also provide performance benefits since they are optimized the first time they are run and then retained in memory for subsequent calls.

The Portal Framework

The portal contains an extensible framework that allows users to build and use individual portal modules to handle the display and management of data.  The following sections will cover the basics of what the portal framework consists of as well as how it was built.

Portal Settings

The portal settings are represented by the PortalSettings Class, which is defined in the Configuration business component.  These settings include the following:

· The Portal ID

· The Portal Name

· The Desktop Tabs Collection

· The Mobile Tabs Collection

· The Currently Active Tab

· The “AlwaysShowEditButton” Setting

The PortalSettings class is updated and placed into the “Context” object upon each web request of the portal application. This is achieved by using the Application_BeginRequest event in the Global.asax file. 

Context.Items.Add("PortalSettings", New PortalSettings(tabIndex, tabId))

Once stored in the Context object, these settings can be obtained from anywhere in the application including all pages, components, and controls by accessing the Context item with the name “PortalSettings”. 

Dim portalSettings As portalSettings = _

CType(HttpContext.Current.Items("PortalSettings"), portalSettings)

Portal Tabs

The Tabs are stored in two public fields of the PortalSettings object we described in the previous section.  The fields, DesktopTabs and MobileTabs, are of the type ArrayList and contain instances of the TabStripDetails class, which represents an individual tab.  Access to the tabs collection is achieved through the PortalSettings Context item. 

Dim portalSettings As portalSettings = _

CType(HttpContext.Current.Items("PortalSettings"), portalSettings)

Dim i As Integer

For i = 0 To portalSettings.DesktopTabs.Count - 1

Dim tab As TabStripDetails = CType(portalSettings.DesktopTabs(i), TabStripDetails)

Next I

The display of the tabs (as shown in Figure 2 below) is handled in the DesktopPortalBanner.ascx user control. The control iterates through the tabs collection, in the same way shown above, checking whether the current user has rights to view the tab.  If the necessary role is met, the tab is added to another collection that will be bound to a DataList. 

Figure 2. Portal Tabs

When a user clicks on a given tab, the PortalSettings object is updated to include the new ActiveTab. When the DesktopPortalBanner.ascx is reloaded, the item in the DataList that corresponds to the active tab’s index is set to the DataList’s SelectedIndex property. 

Portal Modules

Portal Modules provide the actual content of the Portal Starter Kit. The modules are user controls that inherit the PortalModuleControl base class, which provides the necessary communication between the modules and the underlying Portal Framework. The portal comes with eleven built-in portal modules that are available “out of the box,” seven of which are shown in Figure 3. Portal Modules.

Figure 3. Portal Modules

Portal Security

The security design in the portal makes use of both authentication and authorization. Authentication is the process in which the application verifies a user’s identity and credentials. Authorization will actually verify the authenticated user’s permissions for a requested resource. 

The portal supports both forms based and windows based authentication. The authentication mode is defined in the web.config and the User.Identity.Name property maintains the user name. Forms based authentication stores the usernames and passwords in the database and the Windows authentication uses a domain/active directory with the NTLM  challenge/response protocol. The authorization for the portal is handled using role based security to determine whether or not a user has access to a particular resource. Users are grouped into various roles (admins, power users, devs, etc.) and the role mappings are stored in the database. The tabs and modules in the portal maintain access control lists (ACL) to determine who has permission to access the control. This prevents a normal user to access the administration functionality.

For example in the Page_Load event in the admin Tabs.ascx user control, a call is made to IsInRoles(): 

' Verify that the current user has access to access this page

If PortalSecurity.IsInRoles("Admins") = False Then

    Response.Redirect("~/Admin/EditAccessDenied.aspx")

End If

The current users’s role mappings are set for the request in Global.asax in the Application_AuthenticateRequest() event. The Context.User is then set using the GenericPrincipal method and the User.IsInRole can be used to verify whether the current user is in a specific role.

Sub Application_AuthenticateRequest(ByVal sender As Object, _ 

                                    ByVal e As EventArgs)

    If Request.IsAuthenticated = True Then

        Dim roles() As String       

        ' Create the roles cookie if it doesn't exist yet for 

        ' this session.

        If Request.Cookies("portalroles") Is Nothing Then

            ' Get roles from UserRoles table, and add to cookie

            Dim _user As New UsersDB()

            roles = _user.GetRoles(User.Identity.Name)

            ' Create a string to persist the roles

            Dim roleStr As String = ""

            Dim role As String

            For Each role In roles

                roleStr += role

                roleStr += ";"

            Next role

            ' Create a cookie authentication ticket.

            '   version

            '   user name

            '   issue time

            '   expires every hour

            '   don't persist cookie

            '   roles

            Dim ticket As New FormsAuthenticationTicket(1, _

                Context.User.Identity.Name, _

                DateTime.Now, _

                DateTime.Now.AddHours(1), _

                False, _

                roleStr)

            ' Encrypt the ticket

            Dim cookieStr As String = FormsAuthentication.Encrypt(ticket)

            ' Send the cookie to the client

            Response.Cookies("portalroles").Value = cookieStr

            Response.Cookies("portalroles").Path = "/"

            Response.Cookies("portalroles").Expires = _

                DateTime.Now.AddMinutes(1)

        Else

            ' Get roles from roles cookie

            Dim ticket As FormsAuthenticationTicket = _

 FormsAuthentication.Decrypt(Context.Request.Cookies("portalroles").Value)

            'convert the string representation of the role data 

            'into a string array

            Dim userRoles As New ArrayList()

            Dim role As String

            For Each role In ticket.UserData.Split(New Char() {";"c})

                userRoles.Add(role)

            Next role

            roles = CType(userRoles.ToArray(GetType(String)), String())

        End If

        ' Add our own custom principal to the request containing 

        ' the roles in the auth ticket

        Context.User = New GenericPrincipal(Context.User.Identity, roles)

    End If

End Sub

The database calls for all of the role-based checks are contained in Security.vb. 

Administering the ASP.NET Portal Starter Kit

The portal has an online administration tool that allows users in the “Admins” role to manage the security, layout, and content of the portal.  Users that are logged in that belong to the “Admins” role will see an “Admin” tab that takes them to the administration tool. This tool is shown in Figure 4 below.

Figure 4. Portal Administration

The portal administration allows the user to perform a variety of site management and configuration tasks. This is place where new modules can be added, tabs that appear horizontal across the top of the site can be configured, and security roles are defined.

Extending the ASP.NET Portal Starter Kit

The portal was built with the idea of extensibility in mind, providing a way for developers to easily add portal modules that can “plug” into the framework. In this section we will look at the steps that you can follow to build your own portal modules. To do this, we will build a Milestones portal module that will display project milestones.

Extending the Data Layer

Most of the portal modules use the portal database as their primary data store. We will do the same for our example. Therefore the first step is to extend the data layer.  We begin by creating a new Table called Milestones, as shown in Figure 5.

Figure 5.  Milestones Table

Next, we will create the stored procedures required to handle access to the Milestones table. The stored procedures we need are:

· AddMilestone

· DeleteMilestone

· GetMilestone

· GetSingleMilestone

· UpdateMilestone

After the stored procedures are created we need to create a data access layer (DAL) component to provide access to the Milestone procedures. We will define the following methods, the first of which is shown in Listing 1 below. The other methods, elided from this listing for clarity, are in the MilestonesDB.vb file available in the Milestone extension package on the ASP.NET Portal Starter Kit download page.

· ASPNET.StarterKit.Portal.MilestonesDB.GetMilestones()

· ASPNET.StarterKit.Portal.MilestonesDB.GetSingleMilestone()

· ASPNET.StarterKit.Portal.MilestonesDB.DeleteMilestone()

· ASPNET.StarterKit.Portal.MilestonesDB.AddMilestone()

· ASPNET.StarterKit.Portal.MilestonesDB.UpdateMilestone()

Imports System

Imports System.Configuration

Imports System.Data

Imports System.Data.SqlClient

Imports ASPNET.StarterKit.Portal

Namespace ASPNET.StarterKit.Portal

  Public Class MilestonesDB

    Public Function GetMilestones(ByVal moduleId As Integer) As DataSet

      ' Create Instance of Connection and Command Object

      Dim myConnection As _ 

        New SqlConnection(ConfigurationSettings.AppSettings("connectionString"))

      Dim myCommand As New SqlDataAdapter("GetMilestones", myConnection)

      ' Mark the Command as a SPROC

      myCommand.SelectCommand.CommandType = CommandType.StoredProcedure

      ' Add Parameters to SPROC

      Dim parameterModuleId As New SqlParameter("@ModuleId", SqlDbType.Int, 4)

      parameterModuleId.Value = moduleId

      myCommand.SelectCommand.Parameters.Add(parameterModuleId)

      ' Create and Fill the DataSet

      Dim myDataSet As New DataSet()

      myCommand.Fill(myDataSet)

      ' Return the DataSet

      Return myDataSet

    End Function

    ' Other methods elided for clarity.

  End Class

End Namespace

Listing 1
Creating the User Control

After the database is taken care of, the next step is to create the user control that will handle the milestone user interface. The user control will contain a DataGrid that will define three columns: Title, Completion Date, and Status. The implementation of the user control is shown in Listing 2 below. 

<script runat="server">

  Sub Page_Load(ByVal sender As Object, ByVal e As EventArgs)

    Dim milestones As New ASPNET.StarterKit.Portal.MilestonesDB()

    myDataGrid.DataSource = milestones.GetMilestones(ModuleId)

    myDataGrid.DataBind()

  End Sub 

</script>

<asp:DataGrid 

id="myDataGrid" 

HeaderStyle-CssClass="Normal" 

HeaderStyle-Font-Bold="true" 

ItemStyle-CssClass="Normal" 

AutoGenerateColumns="false" 

CellPadding="5" 

Border="0" 

Width="100%" 

EnableViewState="false" 

runat="server">

  <Columns>

    <asp:TemplateColumn>

      <ItemTemplate>

       <asp:HyperLink 

         id="editLink" 

         ImageUrl="~/images/edit.gif" 

         NavigateUrl='<%# "~/DesktopModules/EditMilestones.aspx?ItemID=" & DataBinder.Eval(Container.DataItem,"ItemID") & "&mid=" & ModuleId %>' Visible="<%# IsEditable %>" runat="server" />

      </ItemTemplate>

    </asp:TemplateColumn>

    <asp:BoundColumn DataField="Title" HeaderText="Title" 

                     runat="server" />

    <asp:BoundColumn DataField="EstCompleteDate" 

                     HeaderText="Comp. Date" runat="server" 

                     DataFormatString="{0:d}" />

    <asp:BoundColumn DataField="Status" HeaderText="Status" runat="server" />

  </Columns>

</asp:DataGrid>

Listing 2
Inherit the ASPNET.StarterKit.Portal Module Control Base Class

All portal modules are actually no more than simple ASP.NET user controls that inherit the PortalModuleControl base class. This base class provides all the hooks that are required for the portal module to interact with the framework. To inherit the base class, place the following line at the top of the user control.

<%@ Control language="VB" Inherits=" ASPNET.StarterKit.Portal.PortalModuleControl" %>

Add the Module Title

One of the user controls that are available to portal modules is the Title user control. This control will generate the appropriate HTML.

<portal:title runat="server" />

Add Support for an Edit Page

In order to allow users to edit and add additional Milestones, support for an edit page must be added.   Support is added by passing two additional properties to the Module Title User Control.   

<portal:title 

EditText="Add New Milestone"

EditUrl="~/DesktopModules/EditMilestones.aspx" 

runat="server" />

Create the Edit Page

 Once we have added support for an edit page using the Title User Control, we need to create the actual edit page. In addition to the HTML, we will define four methods:

· Page_Load

· UpdateBtn_Click

· DeleteBtn_Click

· CancelBtn_Click

The source for the DeleteBtn_Click method and the HTML associated with it in the edit page is shown in Listing 3 below. The rest of the implementation may be found in EditMilestones.aspx as part of the Milestone Extension download.

<%@ Page Language="VB" %>

<%@ Register TagPrefix="ASPNETPortal" TagName="Banner" Src="~/DesktopPortalBanner.ascx" %>

<%@ Import Namespace="System.Data.SqlClient" %>

<%@ Import Namespace=" ASPNET.StarterKit.Portal " %>

<script runat="server">

    Private itemId As Integer = 0

    Private moduleId As Integer = 0

    Sub UpdateBtn_Click(ByVal sender As Object, ByVal e As EventArgs)

        ' Only Update if the Entered Data is Valid

        If Page.IsValid = True Then

            ' Create an instance of the Milestone DB component

            Dim milestones As New ASPNET.StarterKit.Portal.MilestonesDB()

            If itemId = 0 Then

                ' Add the milestone within the Milestones table

                milestones.AddMilestone(moduleId, itemId, _






Context.User.Identity.Name, TitleField.Text, _






DateTime.Parse(CompleteDateField.Text), StatusField.Text)

            Else

                ' Update the milestone within the Milestones table

                milestones.UpdateMilestone(moduleId, itemId, _






Context.User.Identity.Name, TitleField.Text, _






DateTime.Parse(CompleteDateField.Text), StatusField.Text)

            End If

            ' Redirect back to the portal home page

            Response.Redirect(CType(ViewState("UrlReferrer"), String))

        End If

    End Sub

</script>

<html>

   <head>

      <link rel="stylesheet" 

         href='<%= Request.ApplicationPath & "/ASPNETPortal.css" %>' type="text/css" />

   </head>

   <body leftmargin="0" bottommargin="0" rightmargin="0" topmargin="0" 

      marginheight="0" marginwidth="0">

      <form runat="server">

         <table width="100%" cellspacing="0" cellpadding="0" border="0">

            <tr>

               <td>

                  <br>

                  <table width="98%" cellspacing="0" cellpadding="4" >

                     <tr valign="top">

                        <td width="100">

                           &nbsp;

                        </td>

                        <td width="*">

                           <p>

                              <asp:LinkButton 

                                 id="deleteButton" 

                                 Text="Delete this item" 

                                 CausesValidation="False" 

                                 runat="server" 

                                 class="CommandButton" 

                                 BorderStyle="none" 

                                 OnClick="DeleteBtn_Click" />

                           </p>

                        </td>

                     </tr>

                  </table>

               </td>

            </tr>

         </table>

      </form>

   </body>

</html>

Listing 3
Adding the New Portal Module to the Framework

The Milestones Portal Module is now complete.  The only step left to perform is to add the module to the portal framework by using the online administrator’s Module Definitions section.  In this section, click on the Add New Module Type to bring up the page shown in Figure 6.  Enter the information for the new module and click Update.   The module can then be added to the different tabs by using the online administrator’s “Tab Name and Layout” section.

Figure 6. Module Type Definition

Conclusion

The portal demonstrates the key techniques used to build a portal web application using ASP.NET. In addition to web-based administration and content management, the portal is also extremely easy to extend as shown in this white paper with the Milestone module. This sample provides a great reference in terms of learning the .NET technologies as well as powerful framework that can be used for Internet or intranet portals.

For More Information

· The complete documentation and source code can be obtained at http://www.asp.net 

· For a live version of the site, please visit http://www.asp.net 

· IBuySpy Listserver http://www.aspfriends.com/aspfriends/aspngibuyspy.asp 

